

SAS script for a macro to apply the recycled prediction method. The script A below is for cost data. For count data, please see Note 1 in the text as well as subsections A1 and A2 below.

A.

```
%macro boot(varname= , rep=);

%let var1 = &varname._post;
%let var2 = &varname._pre;

proc datasets lib=work;
    delete rslt_all_&varname.;
run;

%do i=1 %to &rep;

ods results off;
ods exclude all;
data type.hcu;
set increment (rename=(hcu=hcustatus));
hcu=1;
&var1.=.;
run;
data type.nhcu;
set increment (rename=(hcu=hcustatus));
hcu=0;
&var1.=.;
run;

data type.recycled;
set increment type.hcu type.nhcu;
keep ikn &var1. &var2. hcu hcustatus age sex n_adg lowinc;
run;

proc surveymselect data=type.recycled out=type.boot method=urs
samprate=1 outhits reps=1;
run;

/*NOTE 1: Fitting a two-part model for costs using NL MIXED assuming
gamma distribution. This part is replaced when count data is fit,
assuming either negative binomial or Poisson. For additional code,
please see below: A-NB; B-Poisson*/

proc nlmixed data=type.boot qpoints=1;
parms a0=0 a1=0 a2=0 a3=0 a4=0 a5=0 a6=0
b0=0 b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 log_theta=0;
eta=a0+a1*hcu+a2*&var2.+a3*n_adg+a4*age+a5*sex+a6*lowinc;
exp_eta0=exp(eta);
p0=exp_eta0/(1+exp_eta0);
etah=b0+b1*hcu+b2*&var2.+b3*n_adg+b4*age+b5*sex+b6*lowinc;
mu=exp(etah);
theta=exp(log_theta);
```

```
r=mu/theta;
if &var1.=0 then ll=log(p0);
else ll=log(1-p0)-lgamma(theta)+(theta-1)*log(&var1.)-theta*log(r)-
&var1./r;
model &var1.~general(ll);
predict (1-p0)*mu out=type.pred (keep=ikn hcu hcustatus &var1. pred);
run;

proc means data=type.pred;
where hcustatus ne .;
class hcu; var pred;
output out=type.rslt mean=;
run;

proc transpose data=type.rslt out=type.rslt_tr(rename=(col2=nHCU
col3=HCU) drop=_label_ coll) ;
var pred;
run;

proc append base=rslt_all_&varname. data=type.rslt_tr force; run;

%end;

data type.delta;
set rslt_all_&varname.;
delta=HCU-nHCU;
run;
ods results on;
ods exclude none;

ods html file="/increment/results/incr.&varname..xls";
title "&varname.&rep";
proc univariate data=type.delta;
var delta;
output out=type.pctl_&varname. pctlpre=CI pctlpts=2.5, 97.5;
run;
proc print data=type.pctl_&varname. ;
run;ods html close;

%mend boot;

options symbolgen mprint mcompile ;
%boot(varname=type_cost, rep=1000);
```

```

A1. /*truncated Negative Binomial*/
proc nlmixed data=incr_count;
parms a0=0 a1=0 a2=0 a3=0 a4=0 a5=0 a6=0
b0=0 b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 v=1;
eta0=a0+a1*hcu1+a2*&var2.+a3*n_adg+a4*age+a5*sex1+a6*lowincl;
p0=1/(1+exp(-eta0));

etap=b0+b1*hcu1+b2*&var2.+b3*n_adg+b4*age+b5*sex1+b6*lowincl;
exp_etap=exp(etap);

p=1/(1+(1/v)*exp_etap);
if &var1. =0 then ll=log(p0);
else ll= log(1-p0)+&var1.*log(1-p)-log(p**(-1*(v))-1)+lgamma(&var1.+(v))-lgamma(v)-log(fact(&var1.));
model &var1. ~ general(ll);
predict exp_etap out=admit.pred_trnb (keep=ikn hcu &var1. pred rename = (pred=nb));
ods output parameterestimates=pezph;
ods output fitstatistics=fit1;
run;

A2. /*truncated Poisson*/
proc nlmixed data=incr_count;
parms a0=0 a1=0 a2=0 a3=0 a4=0 a5=0 a6=0
b0=0 b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 ;

eta0=a0+a1*hcu1+a2*&var2.+a3*n_adg+a4*age+a5*sex1+a6*lowincl;
p0=1/(1+exp(-eta0));
etap= b0+b1*hcu1+b2*&var2.+b3*n_adg+b4*age+b5*sex1+b6*lowincl;
exp_etap=exp(etap);

if &var1. =0 then ll=log(p0);
else ll= log(1-p0)-log(1-exp(-exp_etap))-exp_etap-lgamma(&var1.+1)
+&var1.*log(exp_etap);
model &var1. ~ general(ll);
predict exp_etap out=gp.pred_trp (keep=ikn hcu hcustatus &var1. pred rename = (pred=poi));
ods output parameterestimates=pezph;
ods output fitstatistics=fit1;
run;

```